

Simple monadic programming

Implementations in Mathematica / Wolfram Language

Preliminary, unfinished version, 0.4

Anton Antonov
MathematicaForPrediction at WordPress
MathematicaForPrediction at GitHub
MathematicaVsR project at GitHub
June 2017

Introduction

The goal of this document is to provide monad implementations in Mathematica / Wolfram Language (WL) that are both simple to do and useful.

The usefulness of the monadic programming approach comes from several angles:

- 1) easy to construct, read, and modify pipelines of commands,
- 2) easy to program data-polymorphic behaviour,
- 3) easy to program context utilization.

Speaking informally,

Monad programming provides an interface to sequentially structured computations that allows data polymorphic and contextual behavior to be handled by the constructed sequences of operators.

Theoretical and computational background

Monad definition used

My main source for writing this document was Wikipedia article on Monadic programming, [1]. The code in this document is based on the monad

definition given there and in this document that definition is referred to as the “**Haskell definition**”.

Here is the “informal” definition in [1]:

A monad is created by defining a type constructor M and two operations, **bind** and **return** (where **return** is often also called **unit**):

- The unary **return** operation takes a value from a plain type (a) and puts it into a container using the constructor, creating a monadic value (with type $M\ a$).
- The binary **bind** operation “ $>>=$ ” takes as its arguments a monadic value with type $M\ a$ and a function ($a \rightarrow M\ b$) that can transform the value.
 - The bind operator unwraps the plain value with type a embedded in its input monadic value with type $M\ a$, and feeds it to the function.
 - The function then creates a new monadic value, with type $M\ b$, that can be fed to the next bind operators composed in the pipeline.

Note: There are at least two programming monads definitions. The Haskell definition uses “return” and “bind”. Another definition -- used in Scala -- is based the operators “fmap” and “join” (“flatMap”, “map”).

The Haskell definition in Mathematica terms

Here are operators for a monad associated with a certain symbol ‘ M ’:

1. monad **unit** function (“return” in Haskell notation) is `Unit[x_] := M[x];`
2. monad **bind** function (“ $>>=$ ” in Haskell notation) is a rule like `Bind[M[x_], f_] := f[x] with MatchQ[f[x], M[_]] giving True.`

Note that the code definitions in 2 according to the general definition:

- the function `Bind` unwraps the content of `M[_]` and gives it to the function `f`;
- the functions `fi` are responsible to return results wrapped with the monad symbol `M`.

Here is an illustration formula showing a **monad pipeline**:

$$M[data] \xrightarrow{\text{Bind}[M[_], f_1]} f_1 \xrightarrow{\text{Bind}[M[_], f_2]} f_2 \xrightarrow{\text{Bind}[M[_], f_3]} f_3 \xrightarrow{\text{Bind}[M[_], f_4]} \dots \xrightarrow{\text{Bind}[M[_], f_k]} f_k \quad (1)$$

From the definition and formula it should be clear that if for the result `f[x]` of `Bind` the test `MatchQ[f[x], M[_]]` is `True` then the result is ready to be fed to the next binding operation in monad's pipeline. Also, it is easy to program the pipeline functionality with `Fold`:

```
Fold[Bind, M[x], {f1, f2, f3}]
Bind[Bind[Bind[M[x], f1], f2], f3]
```

Premonitions (from the definitions)

Looking at formula (1) we can expect the following points.

- Computations that can be expressed with monad pipelines are easy to construct and read.
- By programming the binding function we can tuck-in certain particular monad behaviours -- this the so called “programmable semicolon” feature of monadic programming.
- Monad pipelines can be constructed with Fold, but with suitable definitions of infix operators like DoubleLongRightArrow (\implies) or NonCommutativeMultiply (**) we can produce code that resembles the pipeline in formula (1).
- A monad pipeline can have a polymorphic behaviour by overloading the signatures of f_i (and if we have to, Bind.)

These points are clarified below.

Programming the Maybe monad

The Maybe monad is a simple example that shows one of the fundamental aspects of monad programming application. (See the description of the monad Maybe in [1].)

The goal of the Maybe monad is to provide easy exception handling in a sequence of chained computational steps. If one of the computation steps fails, then the whole pipeline returns a designated failure symbol, say None; otherwise the result after the last step is wrapped in another designated symbol, say Maybe.

Here is the special version of the generic pipeline formula (1) for the Maybe monad:

$$\text{Maybe}[\text{data}] \xrightarrow{\text{Bind}[m, f_-]} \dots \xrightarrow{\text{Bind}[m, f_-]} \left(\begin{array}{ll} \text{None} & m \equiv \text{None} \\ f_i[x] & m \text{ is Maybe}[x_-] \end{array} \right) \xrightarrow{\text{Bind}[m, f_-]} \dots \quad (2)$$

First definitions

Since we consider the Maybe monad to work on a certain new type, let us define a test for that new type:

```
MaybeUnitQ[x_] := MatchQ[x, None] || MatchQ[x, Maybe[___]];
```

Here is a definition for the binding operation that fits the formula (2):

```
MaybeBind[None, f_] := None;
MaybeBind[Maybe[x_], f_] := Block[{res = f[x]}, If[FreeQ[res, None], res, None]];
```

Note that MaybeBind bails-out more aggressively than prescribed in (2) -- the pipeline goes to None if the current value *has* None, not just when the current value *is* None.

Let us define for completeness the function MaybeUnit which takes an ordinary value and “lifts” it into the monad:

```
MaybeUnit[None] := None;
MaybeUnit[x_] := Maybe[x];
```

Here are a few definitions of monad pipeline functions:

```
MaybeFilter[filterFunc_][xs_] := Maybe@Select[xs, filterFunc[#] &];
MaybeEcho = Maybe@*Echo;
MaybeEchoFunction[f_][x_] := Maybe@EchoFunction[f][x];
```

In order to make the pipeline form of the code we are going to write below let us give definitions to suitable infix operator (like “ \Rightarrow ”) to use MaybeBind:

```
DoubleLongRightArrow[x_?MaybeUnitQ, f_] := MaybeBind[x, f];
DoubleLongRightArrow[x_, y_, z_] := DoubleLongRightArrow[DoubleLongRightArrow[x, y], z];
```

Here is an example of a Maybe monad pipeline using the definitions so far:

```
data = {0.61, 0.48, 0.92, 0.90, 0.32, 0.11};
Maybe[data]⇒ (* lift data into the monad *)
  MaybeFilter[#>0.3&]⇒ (* filter current value *)
  MaybeEcho⇒ (* display current value *)
  (Maybe@Map[If[#<0.4, None, #]&, #]&) (* map values that are too small to None *)
```

```
» {0.61, 0.48, 0.92, 0.9, 0.32}
```

```
None
```

As it was mentioned in the “Premonitions” sub-section the pipeline code of the above example is equivalent to using Fold:

```

Fold[
  MaybeBind,
  Maybe[data],
  {MaybeFilter[#>0.3&],
    MaybeEcho,
    (Maybe@Map[If[#<0.4,None,#]&,#]&)}] (* lift data into the monad *)
  (* filter current value *)
  (* display current value *)
  (* map values that are too small to None *)
» {0.61, 0.48, 0.92, 0.9, 0.32}
None

```

Pipeline flow control definitions

Next it is a good idea to provide some pipeline functions that can control the flow of the data.

Consider the function `MaybeOption[f_]` that would attempts to continue the flow with the result of `f`:

- if `f` produces failure then `MaybeOption` passes current pipeline value;
- if `f` produces a successful result then `MaybeOption` passes that result.

```
MaybeOption[f_][xs_] := Block[{res = f[xs]}, If[FreeQ[res, None], res, Maybe@xs]];
```

Similar simple functions are `MaybeIfElse` and `MaybeWhen`. The function `MaybeIfElse` has three arguments: a test function applied to the current value, a yes-function if the test gives `True`, and no-function if the test gives `False`.

```
MaybeIfElse[testFunc_, fYes_, fNo_][xs_] := Block[{testRes = testFunc[xs]}, If[TrueQ[testRes], fYes[xs], fNo[xs]]];
```

The function `MaybeWhen` is based on `MaybeIfElse` with a no-function that is the monad identity. (See the monad laws below.)

```
MaybeWhen[testFunc_, f_][xs_] := MaybeIfElse[testFunc, f, Maybe][xs];
```

Consider the following example:

```

data={0.61,0.48,0.92,0.90,0.32,0.11};
MaybeUnit[data]⇒
  MaybeFilter[#>0.3&]⇒
  MaybeEcho⇒
  MaybeOption[(Maybe@Map[If[#<0.4,None,#]&,#]&)]⇒
  MaybeEcho⇒
  MaybeWhen[Total[#]>2&,Maybe@Style[#,Red]&]
(* lift data to the monad *)
(* filter-out numbers ≤ 0.3 *)
(* display current value *)
(* attempt mapping values that are < 0.4 *)
(* display current value *)
(* if the total is > 2 color in red *)

```

```

» {0.61, 0.48, 0.92, 0.9, 0.32}
» {0.61, 0.48, 0.92, 0.9, 0.32}
Maybe[{0.61, 0.48, 0.92, 0.9, 0.32}]

```

ToExpression[code]

```

» {0.61, 0.48, 0.92, 0.9, 0.32}
» {0.61, 0.48, 0.92, 0.9, 0.32}
Maybe[{0.61, 0.48, 0.92, 0.9, 0.32}]

```

We can see in the example that `MaybeOption` avoided the failure, and since the total is greater than 2 the pipeline value is colored in red.

There are other control flow functions that can be defined. But that might be more convenience not so much because they are needed. We can always use standard pipeline pure function application to control the flow. Let us demonstrate this with an example.

Consider a control flow function for that applies each of a list of the functions to the current pipeline value and selects a result according to certain scoring function. We can define a function like:

```

MaybeChoice[scoreFunc_, fs_List][xs_] :=
  Block[{res = Through[fs[xs]]}, Maybe@Last[SortBy[res, scoreFunc]]];

```

Consider these data and functions

```

data = {1, 9, 5, 1, 8};
fs = {Sqrt@*Total, Total@*Sqrt, Times@@(Most[#] / Rest[#]) &, Times@@(Rest[#] / Most[#]) &};

```

Then make this kind of pipelines:

```

MaybeUnit[data] ==> MaybeEchoFunction[Through[fs[#]] &] ==> MaybeChoice[N, fs]

```

```

» {2 √6, 5 + 2 √2 + √5, 1/8, 8}
Maybe[5 + 2 √2 + √5]

```

Or directly use a pure function (with the same effect) instead of `MaybeChoice`:

`MaybeUnit[data] \Rightarrow MaybeEchoFunction[Through[fs[#]] &] \Rightarrow (Maybe@Last@SortBy[Through[fs[#]], N] &)`

» $\{2\sqrt{6}, 5 + 2\sqrt{2} + \sqrt{5}, \frac{1}{8}, 8\}$

`Maybe[5 + 2 $\sqrt{2}$ + $\sqrt{5}$]`

It is a monad indeed

Let us convince ourselves that the current definition of `MaybeBind` gives a monad.

Laws

In the monad laws given below “ \Rightarrow ” is for monad’s binding operation and $(x \mapsto \text{expr})$ is for a function in anonymous form.

Here is a table with the laws:

#	name	LHS		RHS
1	Left identity	<code>unit a \Rightarrow f</code>	\equiv	<code>f a</code>
2	Right identity	<code>m \Rightarrow unit</code>	\equiv	<code>m</code>
3	Associativity	<code>(m \Rightarrow f) \Rightarrow g</code>	\equiv	<code>m \Rightarrow (x \mapsto f x \Rightarrow g)</code>

Verification

The verification is straightforward to program and shows that the implemented `Maybe` monad adheres to the monad laws.

#	name	Input	Output
1	Left identity	<code>MaybeUnit[a] \Rightarrow f</code>	<code>f[a]</code>
2	Right identity	<code>Maybe[a] \Rightarrow MaybeUnit</code>	<code>Maybe[a]</code>
3	Associativity LHS	<code>(Maybe[a] \Rightarrow (Maybe[f1[#1]] &)) \Rightarrow (Maybe[f2[#1]] &)</code>	<code>Maybe[f2[f1[a]]]</code>
4	Associativity RHS	<code>Maybe[a] \Rightarrow Function[{x}, Maybe[f1[x]] \Rightarrow (Maybe[f2[#1]] &)]</code>	<code>Maybe[f2[f1[a]]]</code>

Every symbol has a monad with List and Apply

The monad laws are satisfied for every symbol in Mathematica with `List` being the unit and `Apply` being the binding operation.

If we program `Unit` and `DoubleLongRightArrow` like this:

```
Unit = List;
DoubleLongRightArrow[x_, f_] := f @@ x;
```

we can see that monad laws hold:

#	name	Input	Output
1	Left identity	$f @@ \text{Unit}[a]$	$f[a]$
2	Right identity	$\{a\} \Rightarrow \text{Unit}$	$\{a\}$
3	Associativity LHS	$(\{a\} \Rightarrow (\{f1[\#1]\} \&)) \Rightarrow (\{f2[\#1]\} \&)$	$\{f2[f1[a]]\}$
4	Associativity RHS	$\{a\} \Rightarrow \text{Function}[\{x\}, \{f1[x]\} \Rightarrow (\{f2[\#1]\} \&)]$	$\{f2[f1[a]]\}$

Special functions

Let us define several special functions that are useful for certain applications we have in mind (but not of general interest in Maybe monads.)

```
MaybeRandomChoice[n_][xs_] :=
  Maybe@Block[{res = RandomChoice[xs, n]}, If[TrueQ[Head[res] === RandomChoice], None, res]];

MaybeMapToFail[critFunc_][xs_] := MaybeMapToFail[critFunc, 1][xs];
MaybeMapToFail[critFunc_, lvl_][xs_] :=
  If[AtomQ[xs], If[critFunc[xs], None, xs], Maybe@Map[If[critFunc[#], None, #] &, xs, lvl]];

MaybeRandomReal[xs_] := Block[{res = RandomReal[Sequence @@ xs]}, If[NumberQ[res] || ListQ[res], Maybe@res, None]];
```

Here is an example using these functions.

```
MaybeUnit[{{-10, 100}, {30, 4}}] ==>
  MaybeRandomReal ==>
  MaybeEchoFunction[Grid[Quartiles /@ Transpose[#]] &] ==>
  MaybeMapToFail[# < -1 &, {-1}]
17.8388 46.4547 74.2239
13.4514 47.6028 88.0834
» 5.55279 57.3342 79.1495
12.0126 34.2422 65.2693
None
```


The MaybeDivide special function

Since in [1] special attention is given to using the Maybe monad to handle division by 0, here we are going to discuss implementation of such a function.

Here we define the MaybeDivide function to divide the argument with the current pipeline value:

```
MaybeDivide[x_?MaybeUnitQ, y_?MaybeUnitQ] :=
  Block[{yres = MaybeBind[y, MaybeMapToFail[# == 0 &]]},
    If[! FreeQ[yres, None], None, Maybe[x[[1]] / y[[1]]]
  ];
MaybeDivide[y_][xs_] := If[FreeQ[xs, None], MaybeDivide[y, Maybe[xs]], None];
```

This pipeline demonstrates a division success:

```
MaybeUnit[{2, 1, 2}] ==>
  MaybeEcho ==>
  MaybeOption[MaybeDivide[Maybe[{1, 3, 9}]]]
```

» {2, 1, 2}

```
Maybe[{ $\frac{1}{2}$ , 3,  $\frac{9}{2}$ }]
```

This pipeline demonstrates a division failure -- the final result is unchanged because of MaybeOption.

```
MaybeUnit[{0, 2, 1}] ==>
  MaybeEcho ==>
  MaybeOption[MaybeDivide[Maybe[{0, 0, 0}]]]
```

» {0, 2, 1}

```
Maybe[{0, 2, 1}]
```

Polymorphic behavior extensions

Let us extend the Maybe monad developed so far with handling of Dataset objects. This turns out is fairly easy and straightforward.

Here is the formula of the Maybe monad pipeline extended with Dataset objects:

$$M[\text{data}] \xrightarrow{\text{Bind}[m, f_-]} \dots \left(\begin{array}{ll} \text{None} & m \equiv \text{None} \\ f_{i, \text{Dataset}[X]} & m \text{ is Maybe}[\text{Dataset}[X_-]] \\ f_{i, \text{Just}[X]} & m \text{ is Maybe}[X_-] \end{array} \right) \xrightarrow{\text{Bind}[m, f_-]} \dots$$

Maybe monad Dataset extension definitions

The only reason we have to provide a another definition of MaybeBind is because of how FreeQ works over Dataset.

```
ClearAll[MaybeBind]
MaybeBind[Maybe[x_], f_] :=
  Block[{res = f[x]},
    If[MatchQ[res, Maybe[_Dataset]],
      If[FreeQ[res[[1]], None], res, None],
      If[FreeQ[res, None], res, None]
    ]
  ];
```

Next we just provide the Dataset versions of the Maybe monad special definitions (that work with data):

```
MaybeRandomChoice[n_] [xs_Dataset] := Maybe@RandomChoice[xs, n];
MaybeFilter[critFunc_] [xs_Dataset] := Maybe@xs[Select[critFunc]];
MaybeMapToFail[critFunc_] [xs_Dataset] := Maybe@Dataset@Map[If[critFunc[#], None, #] &, xs];
MaybeRandomReal[_Dataset] := Maybe@RandomReal[];
```

Titanic data

Let us load the “Titanic” dataset with the following commands:

```
dataName = "Titanic";
ds = Dataset[Flatten@*List@@@ExampleData[{"MachineLearning", dataName}, "Data"]];
varNames = Flatten[List@@ExampleData[{"MachineLearning", dataName}, "VariableDescriptions"]];
ds = ds[All, AssociationThread[varNames -> #] &];
```

Here is a sample:

`RandomSample[ds, 6]`

passenger class	passenger age	passenger sex	passenger survival
3rd	30.	male	died
3rd	20.	male	survived
3rd	17.	male	died
1st	19.	female	survived
3rd	19.	male	survived
1st	21.	female	survived

Demonstrating polymorphic behavior

In order to demonstrate the polymorphic behavior we going to use the functions `RecordsSummary` and `CrossTabulate` from the package [7].

Dataset

```
SeedRandom[7]
MaybeUnit[ds] =>
  MaybeFilter[#[[2]] > 45 &] =>
    MaybeEchoFunction["\t\tSummary after filtering:" &] =>
      MaybeEchoFunction[RecordsSummary] =>
        MaybeEchoFunction[MatrixForm@CrossTabulate[#[[All, {1, 4}]]] &] =>
          MaybeEchoFunction["\t\tPick 5 rows:" &] =>
            MaybeRandomChoice[5] =>
              MaybeEcho=>
                MaybeEchoFunction["\t\tTest result:" &] =>
                  MaybeEchoFunction[MaybeFilter[#[[4]] == "died" &] =>
                    MaybeMapToFail[#[[4]] == "died" &]
```

» Summary after filtering:

2 passenger age
 1 passenger class
 3 passenger sex
 4 passenger survival

» { 1st 102 , Median 53. , male 103 , died 93 }
 { 2nd 32 , Mean 54.629 female 52 survived 62 }
 { 3rd 21 , 3rd Qu 60. }
 { Max 80. }

» (died survived)
 (1st 50 52)
 (2nd 24 8)
 (3rd 19 2)

» Pick 5 rows:

passenger class	passenger age	passenger sex	passenger survival
2nd	50.	male	died
3rd	49.	male	died
1st	47.	male	died
1st	58.	male	died
1st	54.	male	died

» Test result:

» Maybe [

passenger class	passenger age	passenger sex	passenger survival
2nd	50.	male	died
3rd	49.	male	died
1st	47.	male	died
1st	58.	male	died
1st	54.	male	died

]

None

Array

The same pipeline can be executed over an array instead of a Dataset object. The results are the same, but have a different form in some cases.

```
tbl = Normal[ds[All, Values]];

SeedRandom[7]
MaybeUnit[tbl] =>
  MaybeFilter[#[[2]] > 45 &] =>
    MaybeEchoFunction["\t\tSummary after filtering:" &] =>
      MaybeEchoFunction[RecordsSummary] =>
        MaybeEchoFunction[MatrixForm@CrossTabulate[#[[All, {1, 4}]]] &] =>
          MaybeEchoFunction["\t\tPick 5 rows:" &] =>
            MaybeRandomChoice[5] =>
              MaybeEcho=>
                MaybeEchoFunction["\t\tTest result:" &] =>
                  MaybeEchoFunction[MaybeFilter[#[[4]] == "died" &] =>
                    MaybeMapToFail[#[[4]] == "died" &]
```

```

» Summary after filtering:

      2 column 2
      Min      45.5
      1 column 1 1st Qu 49.      3 column 3      4 column 4
» { 1st 102 , Median 53. , male 103 , died 93 }
    2nd 32 , Mean 54.629 female 52 survived 62
    3rd 21      3rd Qu 60.
      Max      80.

» (
  |died survived|
  |1st  50      52|
  |2nd  24      8 |
  |3rd  19      2 |
)

» Pick 5 rows:
» {{2nd, 50., male, died}, {3rd, 49., male, died}, {1st, 47., male, died}, {1st, 58., male, died}, {1st, 54., male, died}}

» Test result:
» Maybe[{{2nd, 50., male, died}, {3rd, 49., male, died}, {1st, 47., male, died}, {1st, 58., male, died}, {1st, 54., male, died}}]

None

```

Programming and using the State monad

Here we generate the basic functions of a State monad for the symbol “StMon”:

```

Get["~/MathFiles/MathematicaForPrediction/MonadicProgramming/StateMonadCodeGenerator.m"]
Get["~/MathFiles/MathematicaForPrediction/MonadicProgramming/MonadicTracing.m"]

Get["~/MathFiles/MathematicaForPrediction/MathematicaForPredictionUtilities.m"]

GenerateStateMonadCode["StMon", "StringContextNames" → False, "FailureSymbol" → None]

```

```

SeedRandom[232]
Quiet[
  res =
    TraceMonadUnit[StMonUnit[RandomReal[{0, 1}, {3, 2}], <|"color" → Red, "threshold" → 0.5|>]] ⇒
      "(* generate a random matrix and lift into the monad *)" ⇒
      StMonEchoValue ⇒
      (StMon[#1 /. (x_ /; x < #2["threshold"] ⇒ Style[x, #2["color"]]), Join[#2, <|"data" → #1|>]] &) ⇒
      "(* if less that the context threshold: color in red *)" ⇒
      StMonEchoValue ⇒
      StMonModifyContext[Join[#1, <|"color" → Blue, "threshold" → 0.8|>]] ⇒ "(* modify the context *)" ⇒
      (StMon[#2["data"] /. (x_ /; x < #2["threshold"] :> Style[x, #2["color"]]), #2] &) ⇒
      "(* if less that the new context threshold: color in blue *)" ⇒
      StMonEchoValue ⇒
      TraceMonadEchoGrid[Grid[#, Alignment → Left, Dividers → None, Background → Automatic] &];
]
» {{0.0813545, 0.643847}, {0.700647, 0.996779}, {0.861355, 0.287696}}
» {{0.0813545, 0.643847}, {0.700647, 0.996779}, {0.861355, 0.287696}}
» {{0.0813545, 0.643847}, {0.700647, 0.996779}, {0.861355, 0.287696}}

StMonUnit[RandomReal[{0, 1}, {3, 2}], (* generate a random matrix and lift into the monad *)
  Association[color → Red, threshold → 0.5]] ⇒
  StMonEchoValue ⇒
  StMon[#1 /. x_ /; x < #2[threshold] ⇒ Style[x, #2[color]], (* if less that the context threshold: color in red *)
    Join[#2, Association[data → #1]]] & ⇒
» StMonEchoValue ⇒
  StMonModifyContext[ (* modify the context *)
    Join[#1, Association[color → Blue, threshold → 0.8]]] & ⇒
  StMon[#2[data] /. (* if less that the new context threshold: color in blue *)
    x_ /; x < #2[threshold] ⇒ Style[x, #2[color]], #2] & ⇒
  StMonEchoValue

Fold[StMonBind, ReleaseHold[res[[2]]["data"]], res[[2]]["commands"]]

```

```
TraceMonadEchoGrid[GridTableForm[#, TableHeadings → {"code", "comment"}, Dividers → None, Background → Automatic] &]
```

Programming of a dataset manipulation monad

Motivation

Core definitions

```
In[166]:= ClearAll["DMon*"]

In[167]:= DMonUnitQ[x_] := MatchQ[x, None] || MatchQ[x, _Dataset];

In[168]:= DMonUnit[x_] := DMon[x];

In[169]:= DMonBind::ffail = "Fail when applying: `1`";

In[170]:= DMonBind[None, f_] := None;
DMonBind[x_Dataset, f_] :=
  Block[{res = f[x]},
    If[MatchQ[res, _Dataset], res,
      Echo[TemplateApply[StringTemplate[DMonBind::ffail], Inactive[f]]];
      None
    ]
  ];

In[172]:= DMonEcho[None] := None;
DMonEcho[ds_Dataset] := Echo[ds];

In[174]:= DMonEchoFunction[f___][ds_Dataset] := EchoFunction[f][ds];

In[175]:= DMonPrint[_][None] := (Print[None]; None);
DMonPrint[expr_][ds_] := (Print[expr]; ds);

In[177]:= DMonFilter[filterFunc_][ds_Dataset] := ds[Select[filterFunc]];
DMonFilter[___][___] := None;
```



```

In[179]:= DMonSelect[fieldNames : {_String ..}] [ds_Dataset] := ds[All, fieldNames];
          DMonSelect[fieldInds : {_Integer ..}] [ds_Dataset] := ds[All, fieldInds];
          DMonSelect[___] [___] := None;

In[182]:= (*HoldAll[DMonMutate];*)
          DMonMutate[assignments : {Rule[_String, _] ..}] [ds_Dataset] :=
            Block[{t, AList, colNames = Normal[Keys@First@ds]},
              t = With[{r = AList@@assignments, cr = Map[# → # &, colNames]}, Inactive[ds[All, <|cr, r|>]]];
              Activate[t /. RuleDelayed → Rule /. AList → Association]
            ];
          DMonMutate[___] [___] := None;

In[184]:= DMonTransmute[assignments : {Rule[_String, _] ..}] [ds_Dataset] :=
            Block[{t, AList, colNames = Normal[Keys@First@ds]},
              t = With[{r = AList@@assignments}, Inactive[ds[All, r]]];
              Activate[t /. RuleDelayed → Rule /. AList → Association]
            ];
          DMonTransmute[___] [___] := None;

```

Special functions

```

In[186]:= DMonRandomChoice[n_Integer] [ds_Dataset] := RandomChoice[ds, n];
          DMonRandomChoice[___] [___] := None;

In[188]:= DMonRandomSample[n_] [ds_Dataset] := RandomSample[ds, n];
          DMonRandomSample[___] [___] := None;

```

Infix notation

```

In[190]:= DoubleLongRightArrow[x_?DMonUnitQ, f_] := DMonBind[x, f];
          DoubleLongRightArrow[x_, y_, z_] := DoubleLongRightArrow[DoubleLongRightArrow[x, y], z];

```

Data

```
In[192]:= dataName = "Titanic";
ds = Dataset[Flatten@*List@@@ ExampleData[{"MachineLearning", dataName}, "Data"]];
varNames = Flatten[List@@ ExampleData[{"MachineLearning", dataName}, "VariableDescriptions"]];
varNames = StringReplace[varNames, "passenger" ~~ (WhitespaceCharacter ..) → ""];
ds = ds[All, AssociationThread[varNames → #] &];
```

```
In[197]:= ds[[1 ;; 6, All]]
```

Out[197]=

class	age	sex	survival
1st	29.	female	survived
1st	0.9167	male	survived
1st	2.	female	died
1st	30.	male	died
1st	25.	female	died
1st	48.	male	survived

Experiments

In[198]:= **ds** ⇒

```
DMonRandomSample[200] ⇒
DMonFilter[#class != "1st" &] ⇒
DMonSelect[{"age", "survival"}] ⇒
DMonEchoFunction[RecordsSummary];
```

1 age

Missing[] 32

21. 10 2 survival

» { 19. 7 , died 101 }
 { 32. 6 , survived 52 }

30. 5

18. 4

(Other) 89

In[199]:= **ds** ⇒

```
DMonRandomSample[12] ⇒
DMonFilter[#class != "1st" &] ⇒
DMonMutate[{"classSex" -> (#class <> ":" <> #sex &)}] ⇒
DMonSelect[{"classSex", "age", "survival"}] ⇒
DMonEcho ⇒
DMonTransmute[{{"classSex" -> (#classSex &), "sqage" -> ( $\sqrt{\#age}$  &)}]}
```

»

classSex	age	survival
3rd::male	—	died
2nd::male	48.	died
3rd::male	20.	died
2nd::male	26.	died
3rd::female	—	survived
2nd::male	—	died
3rd::female	41.	died
3rd::female	22.	survived
3rd::male	26.	died

Out[199]=


classSex	sqage
3rd::male	$\sqrt{\text{Missing[]}}$
2nd::male	6.9282
3rd::male	4.47214
2nd::male	5.09902
3rd::female	$\sqrt{\text{Missing[]}}$
2nd::male	$\sqrt{\text{Missing[]}}$
3rd::female	6.40312
3rd::female	4.69042
3rd::male	5.09902

Retrospect and summary

How to use monad programming in Mathematica / WL

General work-flow

The following diagram shows my current vision of how to use monadic programming in Mathematica.



Monad code generators and monad completions (packages)

Personal biases

Statements like:

"Monads are the leading design pattern for functional programming."

tend to annoy me -- I see them as a tunnel vision perspective derived from monadic programming.

My biased view on monads is that they are loudly advertised in Haskell, Scala, and F#, but I kind of do not see what is the big deal. Granted, similar things are said about Object-Oriented Programming Design Patterns, but I like them and use them a lot, [2]. In part this article was written to convince myself that monads are not a big deal.

Monad programming is more important for functional programming languages like Haskell and Scala because of their strong type function definition. Not so much in LISP-like languages.

Monad programming style is not so hard to apply/use in LISP or Mathematica. That can be done in several ways. I think the approach presented here is the simplest in WL using "first principles."

References

- [1] Wikipedia entry: Monad (functional programming), URL: [https://en.wikipedia.org/wiki/Monad_\(functional_programming\)](https://en.wikipedia.org/wiki/Monad_(functional_programming)) .
- [2] Anton Antonov, "Implementation of Object-Oriented Programming Design Patterns in Mathematica", (2016) MathematicaForPrediction at GitHub, <https://github.com/antononcube/MathematicaForPrediction>, folder Documentation.
- [3] Anton Antonov, Maybe monad code generator Mathematica package, (2017), MathematicaForPrediction at GitHub.
URL: <https://github.com/antononcube/MathematicaForPrediction/blob/master/MonadicProgramming/MaybeMonadCodeGenerator.m> .
- [4] Anton Antonov, State monad code generator Mathematica package, (2017), MathematicaForPrediction at GitHub.
URL: <https://github.com/antononcube/MathematicaForPrediction/blob/master/MonadicProgramming/StateMonadCodeGenerator.m> .
- [5] Anton Antonov, Monadic contextual classification Mathematica package, (2017), MathematicaForPrediction at GitHub.
URL: <https://github.com/antononcube/MathematicaForPrediction/blob/master/MonadicProgramming/MonadicContextualClassification.m> .
- [6] Anton Antonov, Monadic tracing Mathematica package, (2017), MathematicaForPrediction at GitHub.
URL: <https://github.com/antononcube/MathematicaForPrediction/blob/master/MonadicProgramming/MonadicTracing.m> .
- [7] Anton Antonov, MathematicaForPrediction utilities, (2014), MathematicaForPrediction at GitHub.
URL: <https://github.com/antononcube/MathematicaForPrediction/blob/master/MathematicaForPredictionUtilities.m> .